ISOMETRIES IN THE AFFINE PLANE

Let $E_2 = R^2$ be the euclidean affine plane, with the standard scalar product.
Unless otherwise stated we will work with the standard coordinate system
$R = \{O, B = \{e_1, e_2\}\}$,
which is an orthonormal system. We denoteby (x, y) the coordinates of an arbitrary poin in E_2. We will construct the matrix $M_f(R)$ of an isometry
$f: E_2 \rightarrow E_2$ from the matrices of the next basic issometries :

Translation f of vector v=(a,b):

$$M_f(R)=\begin{bmatrix} 1 & 0 & 0 \\ a & 1 & 0 \\ b & 0 & 1 \end{bmatrix}.$$

Symmetry f:

Axis (line of fixed points)	x=0	y=0
$M_f(R)$	$\begin{bmatrix} 1 & 0 & 0 \\ 0 & -1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$	$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & -1 \end{bmatrix}$

Rotation f of angle α and center the origin.

$$M_f(R)=\begin{bmatrix} 1 & 0 & 0 \\ 0 & \cos(\alpha) & -\sin(\alpha) \\ 0 & \sin(\alpha) & \cos(\alpha) \end{bmatrix}.$$

Exercicies:

Determine the symmetry of axis x+y=1 compound with translation of vector (1,2), that is, obtain its matrix expression.

We have to obtain the matrix
$M_s(R)$ of the symmetry, the matrix $M_t(R)$ of the translation t and multiply them $M_f(R) = M_s(R) M_s(R)$
to obtain the matrix of the composition f=t∘s.

```
> with(plots):
> gvector:=arrow(<1,2>, shape = arrow):
```

```
> gline:=implicitplot(x+y=1,x=-2..2,y=-2..2):
> display(gvector,gline);
```




```
> restart;with(linalg):
> solve(x+y=1);
```

$$\{x = -y + 1, y = y\} \tag{1.1}$$

```
> P:=[1,0]; v:=(1/sqrt(2))*[-1,1];
```

$$P := [1, 0]$$

$$v := \frac{1}{2}\sqrt{2}\,[-1, 1] \tag{1.2}$$

Let R' be the orthonormal coordinate system such that, the line x+y=1 has equation y'=0 in R'.

```
> Rp={P,{v,(1/sqrt(2))*[1,1]}};
```

$$Rp = \left\{[1, 0], \left\{\frac{1}{2}\sqrt{2}\,[-1, 1], \frac{1}{2}\sqrt{2}\,[1, 1]\right\}\right\} \tag{1.3}$$

The matrix of the symmetry of axis y'=0 in R' is M_s (R'):

```
> SRp:=matrix(3,3,[1,0,0,0, 1,0,0,0,-1]);
```

$$SRp := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & -1 \end{bmatrix} \quad (1.4)$$

The matrix of s in R is obtained by the relation:

$$M_s(R) = M(R', R) M_s(R') \; M(R, R')$$

```
> RpR:=matrix(3,3,[1,0,0,1,-1/sqrt(2),1/sqrt(2),0,1/sqrt(2),
  1/sqrt(2)]);RRp:=inverse(RpR);
```

$$RpR := \begin{bmatrix} 1 & 0 & 0 \\ 1 & -\frac{1}{2}\sqrt{2} & \frac{1}{2}\sqrt{2} \\ 0 & \frac{1}{2}\sqrt{2} & \frac{1}{2}\sqrt{2} \end{bmatrix}$$

$$RRp := \begin{bmatrix} 1 & 0 & 0 \\ \frac{1}{2}\sqrt{2} & -\frac{1}{2}\sqrt{2} & \frac{1}{2}\sqrt{2} \\ -\frac{1}{2}\sqrt{2} & \frac{1}{2}\sqrt{2} & \frac{1}{2}\sqrt{2} \end{bmatrix} \quad (1.5)$$

```
> S:=evalm(RpR&*SRp&*RRp);
```

$$S := \begin{bmatrix} 1 & 0 & 0 \\ 1 & 0 & -1 \\ 1 & -1 & 0 \end{bmatrix} \quad (1.6)$$

```
> T:=matrix(3,3,[1,0,0,1,1,0,2,0,1]);
```

$$T := \begin{bmatrix} 1 & 0 & 0 \\ 1 & 1 & 0 \\ 2 & 0 & 1 \end{bmatrix} \quad (1.7)$$

```
> Mf:=evalm(T&*S);
```

$$Mf := \begin{bmatrix} 1 & 0 & 0 \\ 2 & 0 & -1 \\ 3 & -1 & 0 \end{bmatrix} \quad (1.8)$$

Classify the isometry with matrix M and obtain its importants elements.

```
> restart; with(linalg):
> M:=matrix([[1, 0, 0], [2, 0, -1], [3, -1, 0]]);
```

$$M := \begin{bmatrix} 1 & 0 & 0 \\ 2 & 0 & -1 \\ 3 & -1 & 0 \end{bmatrix} \tag{2.1}$$

```
> A:=submatrix(M, 2..3,2..3);b:=submatrix(M,2..3,1..1);id:=diag
  (1,1):rank(A-id);rank(concat(b,A-id));
```

$$A := \begin{bmatrix} 0 & -1 \\ -1 & 0 \end{bmatrix}$$

$$b := \begin{bmatrix} 2 \\ 3 \end{bmatrix}$$

$$1$$

$$2 \tag{2.2}$$

It is a symmetry compound with translation. We compute next the simetry axis and the translation vector.

There are many possibilities but only one with translation vector parallel to the symmetry axis. In this case, the axis is the only invariant line, with direction vector, the eigenvector associated to the eigenvalue 1 (of the linear map associated to the isometry).

```
> eigenvectors(A);
```

$$\left[1, 1, \left\{\left[\begin{array}{cc} -1 & 1 \end{array}\right]\right\}\right], \left[-1, 1, \left\{\left[\begin{array}{cc} 1 & 1 \end{array}\right]\right\}\right] \tag{2.3}$$

```
> X:=matrix(3,1,[1,x,y]): Xf(X):=evalm(M&*X-X);
```

$$Xf(X) := \begin{bmatrix} 0 \\ 2 - y - x \\ 3 - x - y \end{bmatrix} \tag{2.4}$$

The invariant line is orthogonal to the eigenvector associated to the eigenvalue -1. So forcing Xf(X) to be orthogonal to (1,1) we obtain the equation of the invariant line.

```
> print(`Invariant line`);invl:=evalm(matrix(1,3,[1,1,1])&*Xf(X))
  [1,1];
```

Invariant line

$$invl := 5 - 2\,y - 2\,x \tag{2.5}$$

Matrix of the symmetry with such line of fixed points

```
> S:=matrix([[1,0,0],[5/2,0,-1],[5/2,-1,0]]);
```

(2.6)

$$S := \begin{bmatrix} 1 & 0 & 0 \\ \frac{5}{2} & 0 & -1 \\ \frac{5}{2} & -1 & 0 \end{bmatrix} \quad (2.6)$$

Translation ov vector v=Pf(P), with P a point in the invariant line.

```
> solve(invl);
```

$$\left\{x = \frac{5}{2} - y, y = y\right\} \quad (2.7)$$

```
> P:=matrix(3,1,[1,5/2,0]);
```

$$P := \begin{bmatrix} 1 \\ \frac{5}{2} \\ 0 \end{bmatrix} \quad (2.8)$$

```
> evalm(M&*P-P);
```

$$\begin{bmatrix} 0 \\ -\frac{1}{2} \\ \frac{1}{2} \end{bmatrix} \quad (2.9)$$

```
> T:=matrix(3,3,[1,0,0,-1/2,1,0,1/2,0,1]);
```

$$T := \begin{bmatrix} 1 & 0 & 0 \\ -\frac{1}{2} & 1 & 0 \\ \frac{1}{2} & 0 & 1 \end{bmatrix} \quad (2.10)$$

```
> TS:=evalm(T&*S);evalm(S&*T);
```

$$TS := \begin{bmatrix} 1 & 0 & 0 \\ 2 & 0 & -1 \\ 3 & -1 & 0 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 0 & 0 \\ 2 & 0 & -1 \\ 3 & -1 & 0 \end{bmatrix} \quad (2.11)$$

In this case, the composition is commutative but in general is not.

Determine the rotation of angle Π/6 and center C=(1,1).

```
> restart; with(linalg):
> M:=matrix([[1,0,0],[a,cos(Pi/6),-sin(Pi/6)],[b,sin(Pi/6),cos
  (Pi/6)]]);
```

$$M := \begin{bmatrix} 1 & 0 & 0 \\ a & \frac{1}{2}\sqrt{3} & -\frac{1}{2} \\ b & \frac{1}{2} & \frac{1}{2}\sqrt{3} \end{bmatrix} \quad (3.1)$$

```
> A:=submatrix(M,2..3,2..3);
```

$$A := \begin{bmatrix} \frac{1}{2}\sqrt{3} & -\frac{1}{2} \\ \frac{1}{2} & \frac{1}{2}\sqrt{3} \end{bmatrix} \quad (3.2)$$

```
> rank(A-diag(1,1));
Warning, unable to find a provably non-zero pivot
```

$$2 \quad (3.3)$$

We comput and b2 forcing C to be a fixed point.

```
> C:=matrix(3,1,[1,1,1]);
```

$$C := \begin{bmatrix} 1 \\ 1 \\ 1 \end{bmatrix} \quad (3.4)$$

```
> evalm(M&*C-C);
```

$$\begin{bmatrix} 0 \\ a + \frac{1}{2}\sqrt{3} - \frac{3}{2} \\ b - \frac{1}{2} + \frac{1}{2}\sqrt{3} \end{bmatrix} \quad (3.5)$$

```
> sol:=solve({a+1/2*3^(1/2)-3/2, b-1/2+1/2*3^(1/2)},{a,b});
```

$$sol := \left\{a = -\frac{1}{2}\sqrt{3} + \frac{3}{2}, b = \frac{1}{2} - \frac{1}{2}\sqrt{3}\right\} \quad (3.6)$$

```
> assign(sol):[a,b];
```

$$\left[-\frac{1}{2}\sqrt{3} + \frac{3}{2}, \frac{1}{2} - \frac{1}{2}\sqrt{3}\right] \quad (3.7)$$

```
> M=matrix([[1,0,0],[a,cos(Pi/6),-sin(Pi/6)],[b,sin(Pi/6),cos
  (Pi/6)]]);
```

$$M = \begin{bmatrix} 1 & 0 & 0 \\ -\frac{1}{2}\sqrt{3} + \frac{3}{2} & \frac{1}{2}\sqrt{3} & -\frac{1}{2} \\ \frac{1}{2} - \frac{1}{2}\sqrt{3} & \frac{1}{2} & \frac{1}{2}\sqrt{3} \end{bmatrix} \quad (3.8)$$

Observe that Tr(M)=2cos(Pi/6)

```
> alpha:=evalf(arccos(trace(A)/2));
```

$$\alpha := 0.5235987758 \quad (3.9)$$

```
> evalf(Pi/6);
```

$$0.5235987758 \quad (3.10)$$

The rotation of angle Π/6 and center (1,1) is the composition of:
- The rotation of angle Π/6 and,
- the translation of vector (a,b).

```
> G:=matrix([[1,0,0],[0,cos(Pi/6),-sin(Pi/6)],[0,sin(Pi/6),cos
  (Pi/6)]]);
```

$$G := \begin{bmatrix} 1 & 0 & 0 \\ 0 & \frac{1}{2}\sqrt{3} & -\frac{1}{2} \\ 0 & \frac{1}{2} & \frac{1}{2}\sqrt{3} \end{bmatrix} \quad (3.11)$$

```
> T:=matrix([[1,0,0],[a,1,0],[b,0,1]]);
```

$$T := \begin{bmatrix} 1 & 0 & 0 \\ -\frac{1}{2}\sqrt{3} + \frac{3}{2} & 1 & 0 \\ \frac{1}{2} - \frac{1}{2}\sqrt{3} & 0 & 1 \end{bmatrix} \quad (3.12)$$

```
> evalm(T&*G);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ -\frac{1}{2}\sqrt{3} + \frac{3}{2} & \frac{1}{2}\sqrt{3} & -\frac{1}{2} \\ \frac{1}{2} - \frac{1}{2}\sqrt{3} & \frac{1}{2} & \frac{1}{2}\sqrt{3} \end{bmatrix} \quad (3.13)$$

But not conversely.

```
> simplify(evalm(G&*T));
```

(3.14)

$$\begin{bmatrix} 1 & 0 & 0 \\ \sqrt{3}-1 & \frac{1}{2}\sqrt{3} & -\frac{1}{2} \\ 0 & \frac{1}{2} & \frac{1}{2}\sqrt{3} \end{bmatrix} \tag{3.14}$$